
Pole Prvků (poleprvku/poleprvku0.frm)
V této kapitole se seznámíme s tím jak ve VisualBasicu pracovat s poli.

Práce s polem prvků

Vytvoření pole 

První prvek vytvoříme ručně, nastavíme základní vlastnosti a vlastnost Index nastavíme na 0. Ostatní prvky  lze vytvořit kopírováním (potvrdíme dotaz ohledně vytvoření budoucího pole). Prvkům se automaticky nastavuje vlastnost Index.

Co je pole Prvků?

Jsou to řídící prvky stejného typu, které mají stejnou vlastnost (Name) a lišící se vlastností Index PoleTl(0), PoleTl(1) ….

Zjednodušení kódu (poleprvku/poleprvku1.frm)
Po kliknutí na tlačítko budeme chtít aby se objevil prvek MsgBox ve kterém bude stejný nápis jako je nápis na tlačítku na které jsme klikli. Pokud by prvky netvořily pole musela bych napsat tolik procedur, kolik je prvků. Pole prvků obsluhuje jedna událostní procedura, to který prvek je právě obsluhován je ošetřeno pomocí parametru Index této procedury.

Private Sub PoleTl_Click(Index As Integer)

MsgBox PoleTl(Index).Caption

End Sub

Zmizení a zobrazení tlačítek (poleprvku/poleprvku2.frm)
Chceme aby po kliknutí na tlačítko Zmiz všechna tlačítka tvořící pole zmizela. K odstranění tlačítek použijeme cyklus For.

Cyklus For

Cykly se v programech používají vždy tam, kde potřebujete opakovat určitou část programu (tzv. tělo cyklu). Existuje několik druhů cyklů. Nejprve se však naučíte pracovat s tím nejzákladnějším typem cyklu - s cyklem For. Tento cyklus se používá tehdy, když znáte přesný počet opakování příkazů v těle cyklu. 

Syntaxe:

For počítadlo = začátek To konec 
    [ příkazy ]
Next  počítadlo
Do počítadla je před provedením cyklu přiřazena počáteční hodnota začátek. Jestliže je hodnota počítadla menší nebo rovna horní mezi (hodnota konec), provedou se příkazy napsané uvnitř cyklu a zvýší se hodnota počítadla. Cyklus se opakuje dokud hodnota počítadla nepřevýší hodnotu konec.

Příklad: Zmizení pole tlačítek PoleTl

For i = 1 To 5
 PoleTl(i).Visible = False

Next i

Příklad: Zmizení pole tlačítek PoleTl LÉPE
For i = 1 To PoleTl.Count
 PoleTl(i).Visible = False

Next i

Vlastnost Count – celkový počet prvků v zadaném poli

Cyklus For je možné předčasně ukončit (např. po podmínce) pomocí příkazu Exit For.

Řídicí proměnnou cyklu můžeme deklarovat přímo v dané proceduře, bude typu Integer. Proměnné deklarované v těle procedur se nazývají lokální proměnné. Lokální proměnné se deklarují uvnitř procedury, lze se na ně odkazovat jen v dané proceduře (tyto proměnné nelze používat v ostatních částech programu) a po skončení procedury zanikají.

Private Sub cmZmiz_Click()

Dim i As Integer

   For i = 0 To 2

        PoleTl(i).Visible = False

   Next i

End Sub

Cihlová Zeď (zed/zed0.frm)
V některých programech (různé hry) potřebujeme umístit na formulář spoustu ovládacích prvků stejného typu. Pokud bychom se snažili umístit všechny tyto prvky při vytváření formuláře, asi by nám tato činnost trvala velmi dlouho. Někdy ani nevíme kolik prvků budeme potřebovat, protože tento počet závisí na vývoji programu a na přání uživatele. Jakým způsobem se přidávají prvky až za běhu programu si ukážeme na následující jednoduché aplikaci s názvem Zeď. Jak bude přibližně vypadat, vidíte na obrázku.


Popis programu

Po stisknutí tlačítka „Přidej cihlu“ se přidá do zdi další cihla. Když se „dostaví“ jedna řada cihel, začne se stavět další. V jednotlivých řadách jsou cihly posunuty o polovinu jejich délky. Vidíte, že je to velice jednoduché, ale jak to naprogramovat ?

Každá cihla je tvořena ovládacím prvkem Image. Bylo by možné umístit cihly na formulář během vývoje aplikace a nastavit jim vlastnost Visible na hodnotu False. Jednotlivé cihly by nebyly vidět a zobrazovaly by se postupně po stisku tlačítka tím, že by se nastavila vlastnost Visible na hodnotu True. Na první pohled je vidět, že toto řešení je pracné a navíc závislé na velikosti formuláře. Všechny cihly ve zdi jsou vlastně stejné objekty, lišící se pouze ve dvou vlastnostech (Top, Left). Bude tedy jednodušší za běhu programu vytvořit na přání uživatele novou cihlu, která bude vypadat stejně jako ostatní cihly, pouze bude jinak umístěná.

Doposud umíte objekty na formulář umístit v době vývoje aplikace a za běhu s nimi pohybovat nebo měnit jejich velikost. Nyní si ukážeme, jak prvky přidávat až za běhu programu.

Vlastnosti

	Název prvku
	Vlastnost
	Hodnota

	Form1 Form
	Caption
	Zeď

	
	Icon
	zed.ico

	
	MaxButton
	False

	cbpridej CommandButton
	Caption
	Přidej cihlu

	cihla Image
	Index
	0

	
	Picture
	cihla.bmp


Pomocný výklad

Ovládací prvek Image

Zatím jsme na zobrazování obrázků používali prvek PictureBox. Dalším vhodným prvkem pro zobrazování obrázků je ovládací prvek - Image. Někdy se tomuto prvku říká grafické tlačítko. Na prvek PictureBox lze na rozdíl od prvku Image kreslit. Prvek Image se ovládá podobně jako prvek PictureBox. Obrázek se do prvku Image vloží pomocí jeho vlastnosti Picture. Zajímavou vlastností prvku Image je vlastnost - Stretch. Pokud nastavíte její hodnotu na False, obrázek má původní velikost bez ohledu na velikost prvku Image, při hodnotě True bude velikost obrázku závislá na velikosti prvku Image, tzn. že obrázek nemusí vždy zachovávat původní rozměry.

Vkládání prvků za běhu programu

Pokud chcete ovládací prvky vytvářet na formuláři až za běhu programu, nabízí vám Visual Basic velice snadnou metodu. Jednotlivé prvky umístíte do tzv. pole prvků. Všechny prvky budou mít stejné jméno a rozlišovat se budou pomocí vlastnosti Index. Velmi jednoduše pak vyrobíte nový prvek, který se stane součástí tohoto pole. Odkazovat se na jednotlivé prvky budete pomocí jejich jména a indexu.

Příklad: Odkaz na první prvek by mohl vypadat např. takto:


Cihla(1).Visible = True

‘Zviditelnění prvku s indexem 1 z  pole Cihla
První prvek (s indexem 0) vytvoříte při vývoji aplikace a další instance prvku (s indexy 1,2,3…) se budou vytvářet až za běhu programu, ve chvíli kdy bude nový prvek potřeba použít. Události jsou pro všechny objekty v poli společné, ale obslužná procedura má navíc jeden parametr udávající index objektu, který událost vyvolal. Na prvky v poli se odkazujete stejně jako na prvky, které nejsou v poli, jen přidáte závorku s indexem (viz příklad výše uvedený).

Jak tedy konkrétně vytvoříte takové pole prvků?

Na formulář musíte umístit první prvek, ze kterého budete další objekty vytvářet. Nastavíte mu vlastnost Index (např. na hodnotu 0). Další prvky budete vytvářet za běhu programu pomocí příkazu Load.

Syntaxe příkazu: Load objekt
Jako parametr (objekt) uvedete prvek s indexem. Zpravidla se začíná od hodnoty indexu 1 a pokračuje se dále. Nově vzniklý prvek má stejné vlastnosti jako prvek s indexem 0. Pouze vlastnost Visible je vždy u nového prvku nastavena na hodnotu False. Nově vytvořený prvek je tedy neviditelný. Můžete u něho změnit libovolnou vlastnost (pozice, velikost…) a poté objekt zviditelnit nastavením vlastnosti Visible na hodnotu True.

Přiklad: vložení nového prvku Image s indexem 2 a jeho zviditelnění

Load Image(2)
Image(2).Visible = True

Opačným příkazem je příkaz Unload, který uvolní z paměti určený objekt.

Vytvoření řady cihel (zed/zed1.frm)
Pro správné určení polohy cihly použijeme dvě proměnné x a y typu Single. Počáteční poloha bude u levého okraje formuláře ve stejné výšce jako připravený vzor cihla(0). Po klepnutí na tlačítko Přidej cihlu se na formuláři objeví nová cihla.

Option Explicit 'Všechny proměnné musí být deklarovány

Dim x As Single, y As Single
Dim index As Integer
Private Sub cbpridej_Click()

   index = index + 1  'index nové cihličky

   Load cihla(index)  'vytvoříme novou cihličku

'umístění nové cihličky do správné polohy

   cihla(index).Left = x

   cihla(index).Top = y

'nastavení souřadnic do nové polohy

   x = x + cihla(0).Width

'zobrazení cihličky

   cihla(index).Visible = True
End Sub

Private Sub Form_Load()

   x = 0

   y = cihla(0).Top        'počáteční souřadnice první cihly

   index = 0               'index aktuální cihličky

End Sub

Automatické vytvoření řady cihel (zed/zed2.frm)
Místo tlačítka Přidej cihlu si připravíme tlačítko Přidej řadu. Předdefinovaná konstanta N nám udává kolik cihel je v řadě. Pomocí cyklu For přidáme řadu cihel, současně po přidání řady upravíme souřadnice na správné pozice.

Private Sub cbpridej_Click()

Dim i As Integer    'řidicí proměnná cyklu

 For i = 1 To N

   index = index + 1  'index nové cihličky

   Load cihla(index)   'vytvoříme novou cihličku

   'umístění nové cihličky do správné polohy

   cihla(index).Left = x

   cihla(index).Top = y

   'nastavení souřadnic do nové polohy

   x = x + cihla(0).Width

   'zobrazení cihličky

   cihla(index).Visible = True

 Next i

 'příprava souřadnic pro novou řadu

 x = 0

 y = y - cihla(0).Height

End Sub

Vytvoření řady cihel přes celou šířku formuláře (zed/zed3.frm)
V předchozím příkladě jsme tvořili řadu cihliček z předem daného pevného počtu cihel (konstanta N). Nyní bych ráda vytvořila řadu cihliček pře celou šířku formuláře. Máme několik možností, můžeme si dopředu spočítat kolik cihliček v řadě bude a postupovat jako v předchozím případě. Druhou možností je využít opakovaného přidávání cihliček pomocí cyklu omezeného podmínkou. V  našem případě budeme v podmínce testovat, zda se přidávaná cihlička na formulář ještě vejde.

Do … Loop cyklus

Cykly, jak již víte slouží k opakovanému provádění některých částí programu. Obecným cyklem je v jazyce Visual Basic cyklus Do ... Loop. Tento cyklus lze použít v několika variantách podle toho jaká podmínka cyklus ukončuje.

Příkad: Cyklus s podmínkou na začátku provádění cyklu:

Do While podmínka


příkazy

Loop

Příkazy se v tomto cyklu opakují tak dlouho, dokud není splněna podmínka.

Podobně lze vložit podmínku na konec cyklu Do ... Loop While podmínka.

Někdy nemůžeme podmínku umístit ani na začátek, ani na konec cyklu. Pak je možné použít právě obecný cyklus Do ... Loop s tím, že cyklus ukončíme kdekoliv uvnitř cyklu pomocí příkazu Exit Do. Samozřejmě tento příkaz je vhodné umístit do podmíněného příkazu. 

V našem případě budeme v podmínce testovat zda pozice, kam chceme umístit cihličku je menší než šířka formuláře.

Do While x < Form1.ScaleWidth  ‘(nebo pokud chceme celou cihličku}
        ‘x+cihla(0).Width < Form1.ScaleWidth

přidáme cihličku
Loop

Přidání řady cihliček pak bude vypadat následovně:

Private Sub cbpridej_Click()

Do While x + cihla(0).Width < Form1.ScaleWidth

   index = index + 1  'index nové cihličky

   Load cihla(index)   'vytvoříme novou cihličku

   'umístìní nové cihlièky do správné polohy

   cihla(index).Left = x

   cihla(index).Top = y

   'nastavení souřadnic do nové polohy

   x = x + cihla(0).Width

   'zobrazení cihličky

   cihla(index).Visible = True

Loop

 'příprava souřadnic pro novou řadu

 x = 0

 y = y - cihla(0).Height

End Sub

Zbývá vyřešit posun sudých řad o půlku cihly (aby na sebe řady navazovaly). Jedním z možných řešení je vytvořit si pomocnou proměnou typu  Boolean, která si pamatuje zda se jedná o řadu lichou nebo sudou a podle toho nastavuje souřadnici x. (zed/zed4.frm)

